
ビジュアルヘルプ – 図形の描画 

 

描画 

Igor の描画ツールは、ページレイアウトやグラフウィンドウで、ボックス、円、矢印を使って資料を強調表示する

のに役立ちます。 

また、簡単な図の作成にも使うことができます。 

これらの描画ツールはオブジェクト指向型で、出版品質のグラフィックの作成に最適化されています。 

グラフに表示されているウェーブのデータ値をグラフィカルに編集できます。 

データポイントは削除、追加、変更が可能です。 

また、描画するだけで新しいウェーブを作成することもできます。 

コントロールパネルのウィンドウでは、描画ツールを使ってコントロールの背景を装飾することができます。 

Igor の他のすべての機能と同様に、描画ツールも完全にプログラム可能です。 

これにより、プログラマーは、Igor のレパートリーに新しいグラフタイプを追加するコードのパッケージを作るこ

とができます。 

このようなパッケージはプログラマーのみ作成できますが、誰でも利用することができます。 

描画ツールは、ページレイアウト、グラフ、およびコントロールパネルのウィンドウでのみ利用可能です。 

 

ツールパレット 

描画ツールを使うには、まずツールパレットを表示する必要があります。 

グラフやコントロールパネルでは、Graph（グラフ）または Panel（パネル）のメニューにある「Show Tools（ツ

ールを表示）」メニューコマンドで実行します。 

このコマンドは、ウィンドウの左端にツールパレットを追加します。 

ツールパレットは、ページレイアウトウィンドウでは常に表示されます。 

Edit（編集）メニューの「Undo（元に戻す）」「Cut（切り取り）」「Copy（コピー）」「Clear（クリア）」

「Paste（貼り付け）」「Select All（すべて選択）」「Duplicate（複製）」を除くすべての描画コマンドは、ツー

ルパレットに配置されています。 

ツール パレットが表示されると、２つのモードで動作します： 

Operate Mode（操作モード）：一番上の（Operate）ボタンをクリックして、Operate Mode に入ります。この

モードでは、描画ツールパレットは使用できず、ウィンドウは通常通り操作できます。 

Drawing Mode（描画モード）：２番目の（Drawing）ボタンをクリックして描画モードに入ります。すると、ツ

ールパレット全体が表示され、ウィンドウにテキスト、矢印、ボックス、その他の形状を追加できます。 

ツールパレットの下部にある 3つのアイコン -- Drawing Environment アイコン（木と草）、Drawing Layer ア

イコン、および Mover アイコン（ブルドーザー） -- はポップアップメニューです。 

Alt キーを押しながら Mover アイコンをクリックすると、画面外にある描画要素を復元できるポップアップメニュ

ーが表示されます。 

三角形のアイコンをクリックすると、マウスを使ってさまざまな形状を作成するための操作モードを選択できます。 

それのアイコンをクリックして長押しすると、形状を作成するためのダイアログを呼び出すポップアップメニューが

表示されます。 

 



4 

 

 

 

矢印ツール 

１つまたは複数の描画オブジェクトまたはユーザー定義のコントロールを選択、移動、またはサイズを変更するには

矢印ツールを使います。 

オブジェクトの選択、移動、サイズ変更 

図形オブジェクトを１回クリックして選択します。 

選択されると、オブジェクトのサイズを定義する小さな四角形（ハンドル）が表示されます。 

これらのハンドルをドラッグすることで、１つのオブジェクトまたは複数のオブジェクトのサイズを変更で

きます。 

ハンドルをクリックする時に Shiftキーを押しながらクリックすると、カーソルの位置に応じて、オブジェ

クトのサイズ変更を水平方向、垂直方向、または対角線方向に制限できます。 

Alt キーを押す代わりに、カーソルの位置に応じて、リサイズが完全に水平方向または完全に垂直方向に制

限されます。 

オブジェクトをクリックしても選択されない場合は、そのオブジェクトが異なる描画レイヤーにあるか、ま

たは実際には描画オブジェクトではない可能性があります。 

複数の描画オブジェクトを選択するには、追加のオブジェクトを Shiftキーを押しながらクリックします。 

選択からオブジェクトを削除するには、選択中のオブジェクトを Shiftキーを押しながらクリックします。 

複数の描画オブジェクトを選択するには、描画オブジェクトのセットの周囲に選択矩形をドラッグします。 

Alt キーを最初に押さない限り、選択範囲の矩形に完全に囲まれたオブジェクトのみが選択されます。 

Alt キーを押したままにすると、選択矩形に単に接触しているオブジェクトも含まれます。 

選択範囲に追加のオブジェクトを追加するには、選択範囲の矩形をドラッグする前に Shiftキーを押しま

す。 

現在のレイヤー内のすべてのオブジェクトを選択するには、Edit（編集）メニューの「Select All（すべて

を選択）」項目を選択してください。 

 オブジェクトの回転 

選択範囲の表示されているリサイズハンドルから少し離れた場所をクリックすることで、描画オブジェクト

を回転させることができます。 

マウスカーソルが表示されていない回転ハンドル上に移動すると、カーソルの形状が変わり、オブジェクト

を回転させられることを示します。 

その表示されていない回転ハンドルをクリックしてドラッグすると、オブジェクトを回転させることができ

ます。 

 オブジェクトの複製 

Edit（編集）メニューの「Duplicate（複製）」コマンドを使って、選択した描画オブジェクトを複製する

ことができます。 

オブジェクトを複製した直後に、複製したオブジェクトの位置を変更して再度複製すると、新しいオブジェ

クトは、最後に定義した量だけオフセットされます。 

 オブジェクトの削除 

 選択した図形オブジェクトを削除するには、Delete キーを押します。 

 オブジェクトの修正 

図形オブジェクトをダブルクリックすると、そのオブジェクトのダイアログが表示されます。 

ダイアログに表示されている設定を任意に変更できます。 

変更を適用するには、Do It ボタンをクリックします。 

 

シンプルテキストツール 

テキストツールは、通常、同じフォント、サイズ、スタイル、お

よび回転角度で１つのテキスト行を作成するために使われます。 

複数の行を含むフォーマットされたテキストを作成することもで

きます。 

注釈でサポートされているすべてのエスケープコードを使用でき

ます – ヘルプ Annotation を参照してください。 

テキストツールは、コントロールパネルのウィンドウにテキスト

を追加する唯一の手段です。 

テキストの行を作成するには、シンプルテキストアイコンをクリ

ックします。 

カーソルがテキスト入力用の垂直線に変わります。 

次に、ウィンドウ内のテキストの起点にしたい位置にカーソルを

移動し、その位置をクリックして「Create Text（テキストの作

成）」ダイアログを表示します。 

 

一般的なユーザーは、アンカーのポップアップメニューを無視してかまいません。 

これは主に、テキストが数値の原点（X0，Y0）に対して正しく配置されるように、プログラマーが使うものです。 

まず回転が適用され、その後、アンカー設定で指定された原点に対して回転されたテキストが整列されます。 

テキストオブジェクトを編集または変更するには、矢印ツールでオブジェクトをダブルクリックします。 

 

線ツールと矢印ツール 

線ツールを使うと、開始点をクリックし、終了点までドラッグすることで線を描くことができます。 

描画中に Shiftキーを押しながらドラッグすると、線を垂直または水平に制限できます。 

Lines アイコンをクリックしたままにすると、ポップアップメニューが表示され、線を

指定できるダイアログが表示されます。 
 

 

矢印ツールを使って線をダブルクリックすると、同様のダイアロ

グが表示されます。 

このダイアログでは、線のプロパティを変更できます。 

点線パターンのポップアップパレットは、グラフのトレースで使

うものと同じです。 

点線パターンは、Misc（その他）メニューの Dashed Lines（破

線）コマンドを使って調整できます。 

矢印の太さのパラメーターは、矢印の頭部の幅とその長さの比率で

す。 
 

線の太さと矢印の長さのパラメーターはポイント単位で指定され、小数でも指定できます。 

線の開始点と終了点の座標は、選択した座標系に依存します。 

詳細な説明は「図形座標系の設定」を参照してください。 

プログラマーは、線の端が開始座標と終了座標の真ん中に配置される点に注意してください。 

線が非常に太い場合、この点はより明確になります。 

 

長方形、角丸長方形、楕円形 

長方形、角丸長方形、および楕円ツールは、囲む長方形によって定義されるオブジェクトを作成します。 

適切なアイコンをクリックして長押しすると、数値でオブジェクトを指定できるダイアログが表示されます。 

矢印ツールを使ってオブジェクトをダブルクリックすると、同様のダイアログが表示されます。 

オブジェクトをドラッグし始める時に Shiftキーを押したままにすると、正方形または円を作成できます。 

矢印ツールでオブジェクトのサイズを変更している時に Shiftキーを押すと、カーソルの位置が水平、垂直、または

対角線の方向のいずれかに近い場合、その方向に沿ってオブジェクトのサイズが制限されます。 

従って、Shift キーを押しながら対角線方向にドラッグすると、辺の長さが等しく保たれますが、Shift キーを押し

ながら水平方向または垂直方向にドラッグすると、オブジェクトはこれらの方向のどちらか一方のみに沿ってサイズ

が変更されます。 

代わりに Altキーを押しながら対角線方向にドラッグすると、オブジェクトが比例してサイズ変更されます。 

Erase Fill（消去塗りつぶし）モードは、ウィンドウの現在の背景色

で領域を塗りつぶします。 

塗りつぶし背景色は、Solid（単色）または Erase（消去）以外の塗

りつぶしパターンが選択された場合のみ使われます。 

オブジェクトは、その座標によって定義される数学的な長方形の

内側に常に描画されます。 

線の太さに関わらず、このルールは適用されます。 

これは、直線が座標の中心に配置されるのとは異なります。 

角を丸めた長方形の角を調整するには、オブジェクトをダブルク

リックし、表示されるダイアログで RRect Radiusの設定を編集し

ます。 

単位はポイントです。 

 

 

 

 

 

 

 

 

弧と円 

中心と半径で円弧または円を描くには、Oval（楕円）アイコンをクリックしたまま、表

示されるポップアップメニューから Draw Arc（円弧を描画）を選択します。 

クリックし、ドラッグして、中心と半径または開始角度を定義します。 

マウスを動かさずに再度クリックして円を作成するか、移動しながらクリックして弧の

停止角度を定義します。 
 

さまざまなクリック操作とクリック&ドラッグ操作がサポートされており、試行錯誤が推

奨されています。 

弧を編集するには、描画ツールパネルの Oval アイコンをクリックしたまま、表示され

るポップアップメニューから「Edit Arc（弧を編集）」を選択します。 

必要に応じて、弧をクリックして選択します。  

その後、原点、半径、開始角度、終了角度をドラッグできます。 

弧の外観を変更するには、ダブルクリックして「Modify Arc（弧の編集）」ダイアログを表示します。 

楕円とは異なり、弧はポリゴンやベジェ曲線と同様に、現在の点線パターンと矢印先の設定を反映します。 

弧または円の中心は、任意の座標系に存在できます（座標系の描画を参照）が、半径は常に点で表されます。 

 

ポリゴンツール 

ポリゴンツールは、ポリゴンと呼ばれる描画オブジェクトを作成または編集

し、グラフではウェーブを作成または編集することができます。 

詳細については、「ポリゴンとベジエ曲線の描画」を参照してください。 

 

 

 

 

ユーザー形状作成ツール 

このツールは、あなたや他のユーザーが書いたプロシージャコードで定義された描画オ

ブジェクトを作成します。 

詳細については、DrawUserShape コマンドを参照してください。 

例については、エクスペリメント「User Draw Shapes experiment」を開いてくださ

い。  

 

描画環境ポップアップメニュー 

Drawing Environment（描画環境）ポップアップメニューでは、線の太さ、色、塗りつ

ぶしパターン、その他の視覚的な属性を変更できます。 

既存のオブジェクトの属性を変更できます。 

または、まだ作成していないオブジェクトのデフォルトの属性を変更できます。 

既存のオブジェクトの属性を変更するには、まずそのオブジェクトを選択します。 

次に、Environment（環境）ポップアップメニューを使って属性を変更します。  

まだ作成されていないオブジェクトのデフォルト属性を変更するには、オブジェクトが選択されていないことを確認

してください。 

次に、環境ポップアップメニューを使って属性を変更します。 

その時点から、再度変更するまで、新しく作成されるすべてのオブジェクトに新しい属性が適用されます。 

メニュー内の項目は、すべての種類のオブジェクトに影響しません。 

Fill Mode（塗りつぶしモード）と Fill Color（塗りつぶし色）のコマンドは、囲ま

れた形状のみに影響します。 

Line Dash（線破線）と Line Arrow（線矢印）のコマンドは、長方形と楕円には

影響しません。 

Environment ポップアップメニューから All（すべて）を選択するか、オブジェ

クトをダブルクリックして Modify Draw Environment（描画環境の変更）ダイア

ログを呼び出し、複数の属性を変更することができます。 

 

選択ツールで複数のオブジェクトまたはオブジェクトのグループをダブルクリックすると、Modify Draw 

Environment ダイアログが表示されます。 

この場合、表示されるプロパティは最初に選択されたオブジェクトのプロパティですが、プロパティを変更すると、

選択されているすべてのオブジェクトに影響が及びます。 

選択ツールで１つの描画オブジェクトをダブルクリックすると、そのタイプのオブジェクト専用のダイアログが表示

されます。 

 

描画レイヤーポップアップメニュー 

Drawing Layer（描画レイヤー）ポップアップメニューでは、アクティブな描画レイヤ

ーを選択します。 

描画オブジェクトは、アクティブな描画レイヤーでのみ作成および編集できます。 

詳細については、Drawing Layers（描画レイヤー）を参照してください。 

 

 

オブジェクトのポップアップメニュー 

オブジェクトのポップアップメニューでは、さまざまな移動に関する操作を実行できま

す（以降、ムーバーポップアップメニューと呼びます）： 

⚫ 指定されたレイヤー内の描画オブジェクトの前後関係を逆転させる 

⚫ 描画オブジェクトまたはコントロールを互いに、またはグリッドに揃える 

⚫ 描画オブジェクトのグループ化とグループ解除 

⚫ 画面外にある描画オブジェクトまたはコントロールを取得する 

Bring to Front（前面へ移動）、Send to Back（背面へ移動）、Forward（前へ移

動）、Back（後へ移動）コマンドを使って、現在の描画レイヤー内のオブジェクトの前

後関係を調整します。 

 

Align（整列）コマンドは、選択したすべての描画オブジェクトの位置を、最初に選択したオブジェクトを基準に調

整します。 

この機能は、コントロールオブジェクトにも適用されます。 

Distribute（分散）コマンドは、選択したオブジェクトの水平または垂直方向のスペースを均等に調整します。 

このコマンドは、ユーザー定義のパネル内でボタンやその他のコントロールを扱う時に特に便利です。 

この機能は、コントロールだけでなく描画オブジェクトにも適用されます。 

Retrieve（取り出し）コマンドは、画面外にあるオブジェクトを再び表示領域に戻すために使います。 

画面外にあるオブジェクトを取り出すには、ムーバーポップアップメニューの Retrieve（取り出し）サブメニュー

から選択します。 

または、Alt キーを押しながらポップアップメニューからオブジェクトを選択すると、Igor はそのオブジェクトを選

択し、Properties（プロパティ）ダイアログを表示します。 

このオプションを使って、オブジェクトを画面に戻すための数値座標を設定します。 

または、ダイアログをキャンセルし、Delete キーを押してオブジェクトを削除できます。Retrieve は、コントロー

ルおよび描画オブジェクトの両方に適用されます。 

Grid（グリッド）サブメニューでは、グリッドのコントロールに関するオプションが提供されます。 

詳細については「グリッドの描画」を参照してください。 

 

グリッドの描画 

グリッドを表示し、オブジェクトをグリッドにスナップさせるかどうか

を、グリッドの表示/非表示に関係なく指定できます。 

この操作は、ムーバーポップアップメニューの Grid（グリッド）サブメ

ニューで行います。 

デフォルトのグリッドはインチ単位で、８つのサブディビジョンがあり

ます。 
 

グリッドの原点は、ウィンドウまたはサブウィンドウの左上隅です。 

ToolsGrid コマンドを使って、グリッドのプロパティを設定します。 

X グリッドと Y グリッドを独立して指定し、原点、主要グリッド間隔、およびサブディビジョン数を設定できま

す。 

グリッドスナップがオンになっている場合、一時的にオフにするには Caps Lockキーを押してください。 

オブジェクトをドラッグする時、ドラッグを開始したクリック位置に最も近い角がグリッドにスナップされる角にな

ります。 

既存のオブジェクトをグリッドにスナップするには、ムーバーポップアップメニューから Align to Grid（グリッド

に揃える）メニュー項目を選択します。 

 

選択範囲からグリッドを設定 

１つのオブジェクトが選択されている場合、Set Grid from Selection（選択範囲から

グリッドを設定）は、オブジェクトの左上隅をグリッドの原点に設定します。 

2つのオブジェクトが選択されている場合、原点は最初のオブジェクトの左上隅に設定

され、主要なグリッド間隔は 2番目のオブジェクトの左上隅までの距離によって定義さ

れます。 

 

水平方向または垂直方向の差が小さい場合、より大きな間隔によって均一な（Xと Yが等しい）グリッドが定義され

ます。 

そうでない場合、水平方向と垂直方向のグリッドは、それぞれ対応する長さから設定されます。 

 

グリッドスタイル関数 

Style Function（スタイル関数）サブメニューでは、スタイル関数を作成したり、以前に作成したスタイル関数を実

行したりすることができます。 

スタイル関数は、メインのプロシージャウィンドウで「MyGridStyle00」などの名前で作成されます。 

これらの名前は、より意味のある名前に編集することができます。 

描画座標系 

Igor の描画ツールのユニークな機能は、さまざまな座標系を選択できることです。 

オブジェクトごとに、また X 軸と Y 軸で別々に、さまざまな座標系を選択することができます。 

この機能は主にグラフで使われ、ウィンドウのサイズ変更や軸のスケール変更に図形が自動的に適応するように設計

されています。 

オブジェクトのポップアップメニューから、All を選択します。 

座標系は、Modify Draw Environment ダイアログに表示されるポッ

プアップメニューを使って指定します。 

利用可能な座標系は次のとおりです： 

⚫ Absolute（絶対） 

⚫ Relative（相対） 

⚫ Plot Relative（プロット相対） 

⚫ Axis Relative（軸相対） 

⚫ Axis（軸） 
 

 

 

Absolute（絶対） 

Absolute（絶対）モードでは、座標はポイント単位で測定され、コントロールパネルの場合はコントロールパネルの

単位で、ウィンドウの左上隅を基準点として測定されます。 

正の X座標は右方向、正の Y座標は下方向です。 

このモードでは、ウィンドウのサイズ変更によってオブジェクトの位置やサイズが影響を受けません。 

これはページレイアウトやコントロールパネルでのデフォルトかつ推奨されるモードです。 

ウィンドウを縮小すると、一部のオブジェクトがウィンドウの外（画面外）に残り、表示されなくなる可能性があり

ます。 

さらに、絶対座標で指定されたオブジェクトを 1つのウィンドウからコピーし、別の小さなウィンドウに貼り付ける

と、そのオブジェクトは座標が指定する位置に配置されます。 

画面外にある場合でも同様です。 

この現象が発生したと思われる場合は、ムーバーポップアップメニューを使って、画面外にあるオブジェクトをすべ

て取り出すか、迷子になったオブジェクトが見えるまでウィンドウを拡大してください。 

 

Relative（相対） 

このモードでは、座標はウィンドウのサイズに対する割合で測定されます。 

座標値 x=0, y=0 は左上隅を表し、x=1, y=1 は右下隅に対応します。 

ウィンドウのサイズを変更しても、描画オブジェクトが同じ相対位置に保持されるようにしたい場合は、このモード

を使ってください。 

このモードでは、グラフにおいてWYSIWYG（見たままの表示）に近い結果を生成しますが、正確なWYSIWYG結

果ではありません。 

これは、グラフの余白は多くの要因に依存し、ウィンドウサイズには緩やかに依存するためです。 

このモードは、正確な位置合わせが不要なオブジェクト（例えば、トレースの近くから軸の近くを指す矢印など）に

対して良い結果を得られます。 

矢印を特定のデータポイントや軸上の特定の場所に正確に配置したい場合には、この座標系は適していません。 

その場合は、以降の３つの座標系のどれかを使ってください。 

 

Plot Relative（プロット相対）（グラフのみ） 

このシステムは Relativeとほぼ同じですが、ウィンドウの矩形ではなくプロットの矩形を基にしています。 

プロット矩形は、グラフのデフォルトの左軸と下軸、およびそれに対応する鏡像軸で囲まれた矩形です。 

座標 x=0, y=0 は左上隅を表し、x=1, y=1 は右下隅に対応します。 

これはグラフのデフォルト、かつ推奨されるモードです。 

 

プロット相対座標系は、軸に対するサイズと位置を保持する必要があるオブジェクトに最適です。 

良い例として、分割マークで使われる切れ目があります。 

ほとんどのケースでは、以下で説明する、より複雑な Axis-Based（軸ベース（グラフのみ））座標系よりも、Plot 

Relative（プロット相対）または Axis Relative（軸相対）がより適切な選択です。 

 

Axis Relative（軸相対）（グラフのみ） 

この座標系は Plot Relativeとほぼ同じですが、プロット矩形を任意の軸のオフセット分拡張したものを基にしてい

ます。 

軸のオフセットがない場合、結果は Plot Relativeを使った場合と同じになります。 

軸のオフセットについては、Modifying Axes [Axis タブ] で説明しています。 

Axis Relative座標系には、Igor Pro 9.0 以降が必要です。 

Axis Relative座標系は、軸（スタンドオフ領域を含む）に対するサイズと位置を保持する必要があるオブジェクト

に最適です。 

良い例として、展開された矩形の端から始まりまたは終わる背景のハイライト色矩形があり、スタンドオフの隙間を

回避するものです。 

次の図は、それぞれ X範囲が 0から 1の 2つの長方形を示しています。 

上の長方形は Axis Relative を使い、下の長方形は Plot Relative を使っています： 

両方の長方形は、次に説明する軸ベースの座標系を使って、Y座標を指定しています。 

 

Axis-Based（軸ベース）（グラフのみ） 

X 座標系のポップアップメニューには、水平軸のリストが含まれ、Y 座標のポップアップメニューには、垂直軸のリ

ストが含まれます。 

軸座標系を選択すると、その軸に対してプロットされたウェーブデータの場合と同じように、画面上の位置が計算さ

れます。 

ただし、描画オブジェクトの座標はプロット領域に制限されません。 

このモードは、ズームインおよびズームアウトしても、オブジェクトをウェーブの一部に貼り付けたい場合に最適で

す。 

軸は、両方向に無限に延びているものとみなされます。 

このため、軸の範囲が非常に動的であることもあり、画面外にあり表示されないオブジェクトが生成されやすくなり

ます。 

ムーバーポップアップメニューの Retrieve サブメニューを使ってオブジェクトを取り出すことができます。 

また、移動アイコンをクリックする前に Alt キーを押すと、画面外にある各オブジェクトの数値座標を編集すること

ができます。 

オブジェクトが非常に大きくなったり、非常に小さくなったりする可能性もあります。 

軸ベースの描画オブジェクトを使う前に、グラフをほぼ最終形態にしておくことが望ましいです。 

軸ベースの座標系はプログラマーにとって特に興味深いものですが、インタラクティブなタスクにも便利に活用でき

ます。 

例えば、プロットの特定の領域を正確に塗りつぶす長方形を簡単に作成できます。 

軸座標系を使う場合、グラフのサイズを変更したり、軸の範囲を変更したりしても、長方形は正しく表示されます。 

正確な位置にドロップラインやスケール（キャリブレーター）バーを作成することもできます。 

 

描画レイヤー 

レイヤーを使うと、描画オブジェクトの他のウィンドウコンポーネントに対する前面から背面へのレイヤー配置を制

御できます。 

例えば、グラフ内の関心領域を区画化したい場合、グラフのトレースの背後のレイヤーに塗りつぶされた長方形を描

画できます。 

もし、トレースの上のレイヤーに同じ長方形を描画した場合、トレースが隠れます。 

各ウィンドウタイプは、複数の独立した描画レイヤーをサポートしています。 

たとえば、グラフでは、Igor は４組の描画レイヤーを提供しています。 

現在のウィンドウのレイヤー構造を確認し、レイヤーアイコンをクリックすることで別

のレイヤーに切り替えることができます。 

現在のレイヤーはチェックマークで示されています。 

描画レイヤーには名前が付けられています。 

この表は、レイヤーの名前と、各レイヤーをサポートするウィンドウの種類を示してい

ます： 

グラフ  ページレイアウト コントロールパネル 

ProgBack  ProgBack  ProgBack 

UserBack  UserBack  UserBack 

ProgAxes 

UserAxes 

ProgFront ProgFront 

UserFront UserFront 

ProgTop 

UserTop 

Overlay  Overlay  Overlay 
 

描画レイヤーは、ProgSomething と UserSomething という名前のペアで構成されています。 

ユーザーレイヤーは対話的な描画用に、Prog レイヤーは Igor プログラマー用に用意されています。 

この使い方は単なる推奨事項であり、Igor によって強制されるものではありません。 

この推奨事項の目的は、Igor プロシージャが Prog レイヤーに自由にアクセスできるようにすることです。 

Prog レイヤーに描画してから、そのレイヤーを使うプロシージャを実行すると、描画が破損または消去される可能

性があります。 

注釈の上に描画される ProgTop および UserTop 描画レイヤーは、Igor Pro 10.0 で追加されました。 

最上層は Overlay（オーバーレイ）レイヤーです。 

これは、変更が必要な箇所を示すマークアップなどのグラフィック要素を乱すことなく、ユーザーインターフェイス

の描画要素を追加したいプログラマーのために提供されています。 

Overlay レイヤーは、グラフィックの印刷やエクスポートには含まれません。 

このレイヤーは、Igor Pro 7.0 で追加されました。 

現在のレイヤーに描画されたオブジェクトのみを選択できます。 

描画オブジェクトを選択できない場合は、そのオブジェクトが別の描画レイヤーにあるためです。 

適切なレイヤーを見つけるまで、他のレイヤーを試す必要があります。または、ウィンドウの再作成マクロを確認し

てください。 

オブジェクトをレイヤー間で移動するには、切り取り、レイヤーを切り替え、その後貼り付けます。 

または、オブジェクトを右クリックし、Move to Layer（レイヤーに移動）サブメニューから目的のレイヤーを選択

します。 

 

ポリゴンとベジエ曲線の描画 

ポリゴンツールは、ポリゴンと呼ばれる描画オブジェクトを作成または編集し、グラフではウェーブを作成または編

集することができます。 

ポリゴンは、1つ以上の線分（辺）からなる開いた形または閉じた形です。 

ポリゴンは色やパターンで塗りつぶすことができ、始点または終点に矢印の先端を付けることができます。 

開始点と終了点を同じにして閉じたポリゴンを作成することは可能ですが、Igor はそれを閉じた形状として認識し

ません。 

これにより、ポリゴンの開始点または終了点を移動することで、ポリゴンを開くことができます。 

 

新しいポリゴンの作成 

ポリゴンを作成するには、次の 2つの方法があります： 

⚫ セグメントモード：各クリックで新しい頂点が定義されます。 

⚫ フリーハンドモード：滑らかな曲線を掃引するにつれて、新しい頂点追加します。 

セグメントモードでポリゴンを作成するには、ポリゴンアイコンを 1回クリ

ックします。 

次に、ポリゴンの開始位置をクリックします。 

カーソルを移動させると、線分が作成されます。 

２回目のクリックで最初の線分が固定され、2つ目の線分の作成が開始されま

す。 

ポリゴンが完成するまで、線分を書き続けることができます。  

Shift キーを押しながらドラッグすると、移動方向が水平または垂直方向から 15度刻みの角度に制限されます。 

ダブルクリックで最後の頂点を定義したり、最初の頂点をクリックして描画を停止します。 

すると、自動的に編集モードに入り、カーソルが変わり、頂点が四角いハンドルでマークされます。 

(ベジエ曲線では、頂点は「アンカー」と呼ばれます。) 

編集モードでは、ポリゴンを再成形できます。 

編集モードを終了するには、矢印ツールをクリックしてください。 

フリーハンドモードでポリゴンを作成したり、既存のポリゴンを編集したり、ベジエ曲線を描画または編集したりす

るには、ポリゴンアイコンをクリックしたまま、ポップアップメニューが表示されるまで押し続けます。 

次に、次のいずれかの項目を選択します： 

Draw Poly：セグメント化された多角形を作成するモードに入ります。クリックする

と新しいセグメント化された多角形が作成されます。これはアイコンを１回クリックし

た動作と同じです。 

Freehand Poly：フリーハンドのポリゴン作成モードに入ります。クリックするとフ

リーハンドのポリゴンの作成が開始されます。マウスボタンを押したままクリックして

ドラッグすると、マウスボタンを離すまで滑らかな曲線を描けます。マウスボタンを離

すと、自動的に編集モードに入り、形状を変更できます。  

Edit Poly：既存のポリゴンを編集するためのポリゴン編集モードに入ります。詳細な手順は次のセクションで説明

しています。 

Draw Bezier：ベジエポリゴン作成モードに入ります。クリックすると新しいベジエポリゴンが作成されます。ク

リックしてドラッグしてアンカーポイントとコントロールポイントを定義します。最初のポイントをクリックして曲

線を閉じます。 

Edit Bezier：既存のベジエポリゴンを編集するためのベジエポリゴンモードに入ります。詳細な手順は次のセクシ

ョンで説明しています。 

ポリゴンの編集 

編集モードに入るには、ポリゴンアイコンをクリックしたまま、ポップアップメニュー

から Edit Poly（ポリゴンを編集）を選択します。次に、編集したいポリゴンオブジェ

クトをクリックします。 

 

編集モード中は、頂点を移動、追加、削除したり、線分を移動したりできます： 

⚫ 頂点を移動：頂点をクリックしてドラッグすると、その頂点を移動し、関連する辺を伸ばすことができます。 

⚫ 新しい頂点を作成：線分上の頂点の間をクリックしてください。 

⚫ 頂点を削除：Alt キーを押しながら、削除したい頂点をクリックします。 

⚫ 頂点のペアをオフセット：Ctrl キーを押したまま、線分をクリックし、ドラッグします。 

⚫ 閉じたポリゴンを作成：ポリゴンの一方の端の頂点を、もう一方の端の頂点の上にドラッグし、カーソルの右下

に小さな円が表示されるまで移動します。現在、頂点は接続されていて、分離されるまで一緒に移動します。 

接続された頂点は小さな円（通常の頂点は小さな四角形で表されます）で表されます。 

⚫ 閉じたポリゴンの端点を分離：結合した頂点を右クリックし、「Separate 

Polygon Endpoints」を選択します。 

⚫ 直線を描くには、Shift キーを押す：シフトキーを押しながら頂点をドラ

ッグすると、その頂点の位置が、元の頂点の位置を通る水平または垂直の

線、または元の頂点の位置と隣接する 2つの頂点の位置を通る線上に制限

されます。 
 

⚫ スナップグリッドを使う：頂点を垂直方向と水平方向に簡単に揃えることができます。ムーバーツールのポップ

アップメニューをクリックし、Visible（表示）を選択してから Snap（スナップ）を選択すると、グリッドを表

示し、グリッドにスナップします。 

⚫ Ctrl-Z キーを押して直前の操作を元に戻し、Shift-Ctrl-Z キーを押して「再実行」する 

⚫ 編集モードを終了する：矢印ツールをクリックします。 

⚫ ポリゴンの外観を変更：編集モードを終了した後、環境アイコンまたは右クリック

のコンテキストメニューを使って、選択したポリゴンの線サイズや塗りつぶしの色/

パターンなど、その他の属性を調整することができます。ポリゴンの開始点や終了

点に矢印を追加することも可能です。 

矢印ツールでポリゴンをダブルクリックすると、Modify Polygon（ポリゴンの変

更）ダイアログが表示され、同じ変更を行うことができます。 

⚫ Modify Polygon（ポリゴンの変更）ダイアログを使って、曲線の位置やスケール

を変更：X0 Position と Y0 Position は、頂点をクリックしてドラッグするより

も、最初の点の場所をより正確に指定します。 

ダイアログの H スケールおよび V スケールパラメーターを変更して、ポリゴンの

サイズを変更することができます。例えば、両方の設定に 0.5 を入力すると、ポリ

ゴンは通常の半分のサイズに縮小されます。 
 

 

セグメント化されたポリゴン 

NaN の座標ペアを追加することで、ポリゴンをセグメントに分割することができます。 

デフォルトでは、このようなセグメントは、別々のポリゴンとして描画されます。 

このようなポリゴンを透明度の色で塗りつぶすと、重なった部分は２回塗られるため、暗くなります。 

SetDrawEnv subpaths=1 を使うと、セグメントは１つのポリゴン内のサブパスになります。 

セグメントはサブパスを結ぶ線なしで描画されますが、塗りつぶすとポリゴン全体が１つのポリゴンとして扱われる

ため、内部に穴のあるポリゴンを作成することができます。 

これらの穴の埋め方は、SetDrawEnv 関数の fillRule キーワードによって影響を受けます。 

subpaths キーワードは、DrawPoly および DrawBezier で作成されたポリゴンおよびベジエ曲線のみに適用さ

れ、手動で作成されたものには適用されません。 

ポリゴン編集モードに入り、ポリゴンの線分部分を右クリックして Break Line（線を分割）を選択すると、サブパ

スを手動で作成することができます。 

SetDrawEnv subpaths=1 を設定すると、セグメント化されたポリゴンに矢印が追加される方法が変更されます。 

 

ベジエ曲線の編集 

ベジエ曲線を編集するために実行できる操作はいくつかあります： 

アンカーポイントを移動：アンカーポイントをクリックしてドラッグすると、そのアンカーポイントと関連する曲線

を移動または伸縮できます。 

コントロールポイントを移動：コントロールポイントをドラッグします。コント

ロールポイントは、四角形または円形のアンカーポイントに線接続された赤い点

のように表示されます。コントロールポイントに関連するアンカーポイントの反

対側に別のコントロールポイントがある場合、一方のコントロールポイントをド

ラッグすると、両方のコントロールポイントが移動します。  

アンカーポイントに関連付けられた２つのコントロールポイントのうち、１つだけ移動：Alt キーを押したまま、コ

ントロールポイントをクリックしてドラッグすると、関連するアンカーの反対側にあるコントロールポイントとは独

立して移動できます。 

アンカーポイントの真上に位置するコントロールポイントを移動：Ctrl キーを押したまま、コントロールポイント

（およびアンカー）をクリックします。ドラッグすると、アンカーではなくその下のコントロールポイントが移動し

ます。 

新しいアンカーポイントを作成：アンカーポイント間の曲線をクリックします。 

アンカーポイントを削除：Alt キーを押したまま、アンカーポイントをクリックします。 

閉じたベジエ曲線を作成：曲線の一端にあるアンカーを、もう一端のアンカーの上にドラッグし、カーソルの右下に

小さな円が表示されるまで移動します。現在、アンカーは接続されていて、分離されるまで一緒に移動します。 

結合されたアンカーは小さな円（通常のアンカーは小さな四角形で表されます）で表されます。緑色のコントロール

ポイントハンドルは、曲線が閉じていて、一致するコントロールポイントのみを持つコントロールポイントを示しま

す。 

閉じたベジエ曲線の端点を分離：結合されたアンカーを右クリックし、「Separate Bezier Endpoints（ベジェの端

点を分離）」を選択します。 

アンカーポイントのコントロールポイントを変更：アンカーを右クリックすると、アンカーの両側のコントロールポ

イントの長さと角度を設定するコンテキストメニューが表示されます。 

例えば、「Make Sharp Corner（角を鋭くする）」を選択すると、コントロールポイントの長さが 0 に設定され、

アンカーの真上に配置されるため、そのベジエの点は通常のポリゴンと同じになります。編集モードでは、「鋭い

角」は、赤い丸のマーカーが内側に描かれた四角形のマーカーのように見えます。ただし、「鋭く」接続されたアン

カーの表示は、「スムース」接続されたアンカーと同様に、緑色の丸いマーカーのままです。 

コントロールポイントまたは「その他の」コントロールポイントを編集：コントロールポイントをクリックして右ク

リックすると、クリックしたコントロールポイントまたはそのコントロールポイントのアンカーの「反対側」にある

コントロールポイントの長さや角度を設定するコンテキストメニューが表示されます。 

例えば、「Make Other Control Point Parallel（他のコントロールポイントを平行にする）」を選択すると、アンカ

ーポイントの両側でベジェ曲線が接線となるように、他の制御点の角度が設定されます。この角度は、クリックした

制御点によって既に定義されている角度に基づきます。 

２つのアンカー間の曲線を定義するコントロールポイントを修正：アンカー間の曲線を右クリックすると、その曲線

の開始点と終了点にある２つのアンカーに付随する２つのコントロールポイントの長さと角度を設定するコンテキス

トメニューが表示されます。 

例えば、「Make 90 Degree Arc（90 度の円弧を作成）」を選択すると、２つのコントロールポイントの角度と長

さが設定され、２つのアンカーポイントの間に 90度の円弧が描かれます。この選択肢を利用可能にするためには、

２つのアンカー間の直線に曲がりが必要であり、これにより弧の正しい方向を選択できるようになります。 

「Break Curve to Start Another Bezier（曲線を分割して別のベジエ曲線を開始）」を選択すると、アンカー間の 

２つのコントロールポイントが１つの (NaN,NaN) 座標ペアに置き換えられ、２つの別々のベジエ曲線が描画され

ます。 

セグメント化されたベジェ曲線の詳細は下記を参照してください。 「Remove All Bezier Breaks（すべてのベジェ

の分割点を削除）」を選択すると、ベジェオブジェクトからすべての (NaN, NaN) 座標ペアが削除されます。 

直線を描くには、Shift キーを押す：Shift キーを押しながらベジェコントロールハンドルをドラッグすると、移動

方向が水平/垂直方向から 15度刻みの角度に制限されます。 

ポリゴンまたはベジエのアンカーポイントをドラッグしながら Shift キーを押すと、アンカーの位置が、元のアンカ

ー位置と２つの隣接するアンカーポイントの位置を通る最も近い線にスナップします。Shift キーを押しながらドラ

ッグすると、移動方向が水平または垂直方向に制限されます。 

スナップグリッドを使う：頂点を垂直方向と水平方向に簡単に揃えます。「ムーバー」ツールのポップアップメニュ

ーをクリックし、「Visible（表示）」を選択してから「Snap（スナップ）」を選択して、グリッドを表示し、グリ

ッドにスナップさせます。 

Ctrl-Z キーを押して直前の操作を元に戻し、Shift-Ctrl-Z キーを押して「再実行」する 

編集モードを終了する：矢印ツールをクリックします。 

ベジエ曲線の外観を変更：編集モードを終了した後、環境アイコンまたは右クリックのコンテキストメニューを使っ

て、選択したポリゴンまたはベジエ曲線の他の属性（線のサイズ、塗りつぶしの色やパターンなど）を調整すること

ができます。ベジエ曲線の開始点や終了点に矢印を追加することもできます。 

ベジエ曲線をダブルクリックすると、同じ変更を加えるための Modify Bezier（ベジエ曲線の編集）ダイアログを呼

び出すことができます。 

曲線を移動または拡大縮小する：Modify Bezier（ベジエ曲線の編集）ダイアログを使います。 

X0と Y0の位置設定は、曲線をドラッグして移動させるよりも、最初の点の場所をより正確に指定します。 

ダイアログの H スケールおよび V スケールパラメーターを変更すると、ベジエ曲線のサイズを変更できます。例

えば、両方の設定に 0.5 を入力すると、ベジエ曲線が通常の半分のサイズに縮小されます。 

 

セグメント化されたベジエ曲線 

ベジエ曲線を、NaNの座標ペアを追加することでセグメントに分割することができます。 

デフォルトでは、このようなセグメントは、別々のベジエ曲線として描画されます。 

このような曲線に透明度のある色を塗りつぶすと、重なり合った部分は 2回塗りつぶされるため、暗くなります。 

SetDrawEnv subpaths=1 を指定した場合、セグメントは単一のベジェ曲線内のサブパスとして扱われます。 

セグメントはサブパス同士を接続する線なしで描かれますが、塗りつぶすと全体が１つのベジェ曲線として扱われる

ため、内部に穴のある曲線を作成することが可能です。 

これらの穴の埋め方は、SetDrawEnv 関数の fillRule キーワードによって影響を受けます。 

subpaths キーワードは、DrawPoly および DrawBezier で作成されたポリゴンおよびベジエ曲線のみに適用さ

れ、手動で作成されたものには適用されません。 

サブパスを手動で作成するには、編集ベジエモードに切り替えて、曲線を右クリックし、Break Curve（曲線を分

割）を選択します。 

SetDrawEnv subpaths=1 に設定する副作用として、セグメント化されたベジェ曲線に矢印が追加される方法が変

更されます。 

SetDrawEnv subpaths=0（デフォルト）の場合、各セグメントに矢印が追加され、それぞれが独立したベジェ曲線

として扱われます。 

subpaths=1 を指定すると、矢印はベジェ曲線全体における最初の点または最後の点にのみ追加されます。 

次のデモでは、6時位置を起点と終点とし、3時、12時、9時位置にアンカーポイントを設定した 5つのアンカーポ

イントを使用して、2つの円近似からドーナツを作成する方法を示しています。 

NaN、NaNのペアが 2つのセグメントを区切ります。 

Function DemoDrawBezierAndSubpath() 

 Display as "Bezier Curves With Subpaths" 

 ModifyGraph width={Aspect,1}, margin=18, expand=2 

 // 比較のため、下に円を描く 

 SetDrawLayer/K UserBack 

 // プロットの相対座標を設定（0～1）、塗りつぶしなし 

 SetDrawEnv xcoord=prel, ycoord=prel, fillpat= 0, save 

 SetDrawEnv linethick= 3, linefgc= (32792,65535,1) // 広い薄い緑の線 

 DrawOval 0,0,1,1 

 // コントロールポイントの長さ = 0.55.. * 90 度のセグメントで構成される円の半径 

 Variable radius= 0.5 

 Variable len= 0.55191502449 * radius 

 Variable xc= 0.5, yc= 0.5   // 円の原点 

 // 4つの 90度のベジエ曲線セグメントを計算するには、5つのアンカーが必要 

 Variable anchor0x, anchor0y, t0x, t0y 

 Variable l1x, l1y, anchor1x, anchor1y, t1x, t1y 

 Variable l2x, l2y, anchor2x, anchor2y, t2x, t2y 

 Variable l3x, l3y, anchor3x, anchor3y, t3x, t3y 

 Variable l4x, l4y, anchor4x, anchor4y 

 // 開始アンカーポイントには後続のコントロールポイントのみ持つ 

 anchor0x= xc; anchor0y=yc+radius  // 開始点は 6時方向 

 t0x= anchor0x+len; t0y= anchor0y  // 後続コントロールポイント 

 // 2番目のアンカーポイントには、前方と後続の両方にコントロールポイントがある 

 l1x=xc+radius; l1y = yc+len   // 前方のコントロールポイント 

 anchor1x= xc+radius; anchor1y= yc  // 3時方向のアンカー 

 t1x=l1x; t1y = yc-len    // 後続コントロールポイント 

 // 3番目のアンカーポイントには、前方と後続の両方にコントロールポイントがある 

 l2x=xc+len; l2y = yc-radius   // 前方のコントロールポイント 

 anchor2x= xc; anchor2y= yc-radius  // 12時方向のアンカー 

 t2x=xc-len; t2y = l2y    // 後続コントロールポイント 

 // 4番目のアンカーポイントには、前方と後続の両方にコントロールポイントがある 

 l3x=xc-radius; l3y = yc-len   // 前方のコントロールポイント 

 anchor3x= xc-radius; anchor3y= yc  // 9時方向のアンカー 

 t3x=l3x; t3y = anchor3y+len   // 後続コントロールポイント 

 // 最後（5番目）のアンカーポイントには前方のコントロールポイントのみ持つ 

 l4x=xc-len; l4y = yc+radius   // 前方のコントロールポイント 

 anchor4x= xc; anchor4y= l4y   // 9時方向、最初のアンカーと同じ 

 // 参照円の上部にベジエ曲線を描画する 

 SetDrawLayer/K UserFront 

 SetDrawEnv xcoord=prel, ycoord=prel, subpaths=1, fillRule=1, save 

 SetDrawEnv linethick= 1, linefgc= (65535,0,0)   // 赤い線 

 SetDrawEnv fillfgc= (16385,28398,65535,49151),fillpat= 1 // 透明の青で塗る 

 // 外の円を描く 

 // 明確にするため、アンカーごとに 1つのコマンドを指定 

 DrawBezier anchor0x, anchor0y, 1,1, {anchor0x, anchor0y, t0x, t0y} 

 DrawBezier/A {l1x, l1y, anchor1x, anchor1y, t1x, t1y} 

 DrawBezier/A {l2x, l2y, anchor2x, anchor2y, t2x, t2y} 

 DrawBezier/A {l3x, l3y, anchor3x, anchor3y, t3x, t3y} 

 DrawBezier/A {l4x, l4y, anchor4x, anchor4y} 

 // NaN,NaN でサブパスの分割ポイントを追加 

 DrawBezier/A {NaN, NaN} 

 // （サブパスとして）より小さな半径の円を追加 

 radius= 0.25 

 len= 0.55191502449 * radius 

 anchor0x= xc; anchor0y=yc+radius  // 6時方向のアンカー 

 t0x= anchor0x+len; t0y= anchor0y 

 l1x=xc+radius; l1y = yc+len 

 anchor1x= xc+radius; anchor1y= yc  // 3時方向のアンカー 

 t1x=l1x; t1y = yc-len 

 l2x=xc+len; l2y = yc-radius 

 anchor2x= xc; anchor2y= yc-radius  // 12時方向のアンカー 

 t2x=xc-len; t2y = l2y 

 l3x=xc-radius; l3y = yc-len 

 anchor3x= xc-radius; anchor3y= yc  // 9時方向のアンカー 

 t3x=l3x; t3y = anchor3y+len 

 l4x=xc-len; l4y = yc+radius 

 anchor4x= xc; anchor4y= l4y   // 6時方向、最初のアンカーと同じ 

 DrawBezier/A {anchor0x, anchor0y, t0x, t0y} 

 DrawBezier/A {l1x, l1y, anchor1x, anchor1y, t1x, t1y} 

 DrawBezier/A {l2x, l2y, anchor2x, anchor2y, t2x, t2y} 

 DrawBezier/A {l3x, l3y, anchor3x, anchor3y, t3x, t3y} 

 DrawBezier/A {l4x, l4y, anchor4x, anchor4y} 

 // ウェーブを使って半円を描く 

 Make/O bezierX= {anchor0x, t0x, l1x, anchor1x, t1x, l2x, anchor2x } 

 Make/O bezierY= {anchor0y, t0y, l1y, anchor1y, t1y, l2y, anchor2y } 

 SetDrawEnv linethick= 2, linefgc= (0,0,0)  // 線を切断 

 DrawBezier anchor0x, anchor0y, 1,1, bezierX, bezierY 

End 

ポリゴンとベジエ曲線の塗りつぶしルール 

交差するエッジのない単純なポリゴンおよびベジエ曲線は、明確に塗りつぶされます。 

形状内の点は、色で塗りつぶされます。 

しかし、形状に交差する辺がある場合、何が内部で何が外部かをどのように判断するのでしょうか？ 

一般的に使われるルールは、偶数・奇数ルールと非ゼロ巻きルールという２つがあります。 

Igor は、デフォルトで巻きルールを使っています。 

指定されたポイントから始まり、無限遠まで直線を引きます。 

辺が交差した場合、その辺が線の上から下（または線の始点に対して時計回り）に引かれている場合、1を引きま

す。 

線分が下から上（または反時計回り）に描かれている場合、1を加えます。 

結果が 0でない場合、その領域は内部にあり、着色する必要があります。 

SetDrawEnv fillRule=1 を使って、偶数・奇数ルールを指定することができます。 

fillRule キーワードは、DrawPoly および DrawBezier で作成されたポリゴンおよびベジエ曲線のみに適用され、

手動で作成されたものには適用されません。 

手動で作成したポリゴンおよびベジエ曲線は、デフォルトでは巻き塗りつぶしルールに従います。 

このルールは、Modify Polygon（ポリゴンの変更）または Modify Bezier（ベジエの変更）ダイアログで変更でき

ます。 

 

あるポイントから始まり、無限遠まで直線を引きます。 

その直線が交差する辺の数を数えます。 

結果が奇数であれば、そのポイントは内側です。 

偶数であれば、外側です。 

 

ウェーブの描画と編集 

グラフでは、ポリゴンツールを使って、ポリゴンの描画で説明したのと同じ方法でウェーブを作成または編集するこ

とができます。 

ポリゴンツールをクリックして押し続けます。 

Igor は、通常のポリゴンおよびベジエコマンドに加え、以下のウェーブ描画コマンドを含むポップアップメニュー

を表示します。 

Draw Wave（ウェーブを描画） 

 Draw Wave Monotonic（単調増加ウェーブを描画） 

 Draw Freehand Wave（フリーハンドウェーブを描画） 

 Draw Freehand Wave Monotonic（単調増加フリーハンドウェーブを描画） 

 Edit Wave（ウェーブを編集） 

 Edit Wave Monotonic（単調増加ウェーブを編集） 

最初の４つのコマンドは、W_XPolynn および W_YPolynn という形式の名前を持

つ２つのウェーブを作成して追加します。 

ここで、nn は、名前が一意になるように選択した数字です。 

ウェーブの描画とフリーハンドウェーブの描画は、上記の対応するポリゴンの描画と

まったく同じように機能します。 

単調なバリエーションでは、X 方向へのバックトラックはできません。 

ポリゴンと同様、描画が終了すると編集モードになります。 

既存のウェーブ、または XY ペアとして表示されているウェーブのペアは、Edit 

Wave（ウェーブを編集）コマンドのどちらかを選択し、編集するウェーブのトレー

スをクリックして編集できます。 
 

ここでも、単調なバリエーションは X 方向へのバックトラックを防ぎます。 

XY モードで表示されていないウェーブを編集した場合、X 座標はポイント番号から計算されるため、X 座標を調整

することはできません。 

「プログラムによる描画」で説明している GraphWaveDraw および GraphWaveEdit コマンドを使って、プログ

ラムでウェーブの描画または編集を開始することができます。 

 

描画のエクスポートとインポート 

Igor 内でのコピー／ペースト 

Edit（編集）メニューを使うと、図形オブジェクトをカット、コピー、クリア、ペーストすることができます。 

これは、予想されている通りに行えます。 

描画したオブジェクトは、他のプログラムによって変更されない限り、Igor のプロパティをすべて保持します。 

ただし、Igor の図面を別のプログラムにエクスポートしてから Igor にコピーし直した場合、図面に変更を加えて

いなくても、その図面は Igor で編集できなくなります。 

選択した描画オブジェクトをクリップボードにコピーし、その後 Igor にペーストすると、その座標が保持されま

す。 

ただし、この操作により、ペーストされたオブジェクトが座標系の表示領域外に配置される可能性があります。 

これは、オブジェクトの座標が表示領域内に含まれていない場合です。 

ペーストが期待した結果にならない場合は、一部のオブジェクトが画面外にペーストされた可能性があります。 

画面外にペーストされたオブジェクトを確認または取得するには、ムーバーアイコンを使用できます。 

 

描画レイヤーに画像をペースト 

描画プログラムから画像をペーストする方法は、予想と異なる動作をする場合があります。 

Igor は、コンポーネントオブジェクトにアクセスできるように、画像を分解しようとはしません。 

Igor は画像全体を 1 つのオブジェクトとして扱い、移動やサイズ変更はできますが、それ以外の調整はできませ

ん。 

ペーストした画像のスケールを変更するには、オブジェクトを選択した状態でハンドルをドラッグするか、オブジェ

クトをダブルクリックして表示されるダイアログで xと yのスケール係数を設定します。 

プログラムによる描画 

Igor のすべての描画機能は、Igor プロシージャから使うことができます。 

このセクションでは、描画プログラミングについて一般的な説明と、使用方法に関する戦略、およびサンプルコード

について説明します。 

Igor の描画のプログラム可能な性質は、新しいグラフタイプを作成する時に特に役立ちます。 

例えば、Igor は極座標グラフをネイティブのグラフタイプとしてサポートしていませんが、高品質の極座標グラフ

を作成する極座標グラフパッケージを作成することができます。 

（このパッケージのデモを見るには、File → Example Experiments → Graphing Techniques → New Polar 

Graph Demo を選択してください。） 

プログラマーではない方はパッケージをそのまま使用できます。 

一方、プログラマーはコードを自分の目的に合わせて修正したり、自分のプロジェクトに役立つコードのスニペット

を抽出したりできます。 

対話的に描画を行い、次に Igor にウィンドウの再作成マクロの作成を実行することで、描画プログラミングプロジ

ェクトをすばやく開始することができます(閉じるボタンをクリックし、再作成マクロを保存して、Procedure ウィ

ンドウを確認してください)。 

描画プログラミングのプロジェクトを素早く開始するには、まずプロジェクトの基盤となるコードをコピーして貼り

付けます。 

その後、プロジェクトに役立つコードのスニペットを抽出できます。 

多くの場合、文字列で指定された座標値を計算結果に置き換えるだけで、すぐにプロジェクトを動作させることがで

きます。 

 

描画コマンド 

以下は、描画に関連するコマンドの一覧です： 

DrawAction  DrawArc  DrawBezier  DrawLine 

DrawOval  DrawPICT  DrawPoly  DrawRect 

DrawRRect  DrawText  DrawUserShape  GraphNormal 

GraphWaveDraw  GraphWaveEdit  HideTools  SetDashPattern 

SetDrawEnv  SetDrawLayer  ShowTools  ToolsGrid 

 

SetDrawLayerコマンド 

SetDrawLayer を使って、以降の描画コマンドが影響を与えるレイヤーを指定します。 

/K フラグを使うと、指定された描画レイヤーの現在の内容が削除されます。 

SetDrawLayer と /K フラグの使用に関する考慮事項については、「描画プログラミングの戦略」を参照してくだ

さい。 

 

SetDrawEnvコマンド 

これは描画機能の主要なコマンドです。 

１つのオブジェクトの特性を指定するため、将来のオブジェクトのデフォルト描画環境を指定するため、およびオブ

ジェクトのグループを作成するために使われます。 

SetDrawEnv コマンドを複数回連続して実行できます。 

その効果は累積されます。 

デフォルトでは、SetDrawEnv コマンドのグループは次の描画コマンドのみに影響を及ぼします。 

最初の描画コマンドに続く描画コマンドは、SetDrawEnv コマンドが発行される前に有効だったデフォルト設定を

使います。 

例えば、これらの SetDrawEnvコマンドは、２つの DrawTextコマンドのうち最初の１つに対してのみフォントと

フォントサイズを変更します： 

SetDrawEnv fname="Courier New" 

SetDrawEnv fsize=18   // 18 point Courier New, commands accumulate 

DrawText 0,1,"This is in 18 point Courier New" 

DrawText 0,0,"Has font and size in use before SetDrawEnv commands" 

 

SetDrawEnv で save キーワードを使うと、設定を永続化できます。 

save キーワードの一般的な使用方法は、一連の SetDrawEnv コマンドの最後のコマンドの末尾に指定することで

す。 

永続設定を使うと、オブジェクトごとに SetDrawEnvコマンドを再実行する必要なく、同じ特性を持つ複数のオブ

ジェクトを描画できます。 

オブジェクトのグループを作るには、gstart と gstop キーワードを使って SetDrawEnv コマンドで描画コマンド

のグループを囲みます。 

グループ化は、純粋にユーザーインターフェイスの概念です。 

オブジェクトは、グループ化に関係なくまったく同じように描画されます。 

 

Draw<object>コマンド 

DrawPoly および DrawBezier コマンドには、次の２つの種類があります： 

⚫ リテラル：頂点またはコントロールポイントを、一連の文字列で指定することができます。（対話的に作成され

たポリゴンおよびベジエ曲線は、常に文字列のタイプです） 

⚫ ウェーブ：ウェーブを使って、頂点またはコントロールポイントを定義することができます。 

ポリゴンおよびベジエ曲線は長さに制限がないため、/A フラグを使うと、オブジェクトの定義を複数行にまたがっ

て記述することができます。 

１つのポイントのみでポリゴンを指定することもできます。 

これを使って、原点頂点に頂点を追加するループを設定します。 

頂点を追加せずにポリゴンを１つの頂点のままにしておくと、ユーザーはポリゴンを表示したり選択したりできなく

なることに注意してください。 

 リテラル vs ウェーブ 

プログラマーは、ポリゴンやベジエ曲線を作成する時に、リテラル型とウェーブ型のどちらかのポリゴンタ

イプを選択する必要があります。 

このセクションでは、2つのタイプの違いについて説明します。 

リテラル方式の利点は、ユーザーを混乱させるような多数のウェーブで Experimentを乱雑にしないことで

す。 

また、このようなオブジェクトはすべて互いに独立しているという利点もあります。 

ウェーブ方式では、オブジェクトは独立していません。 

ユーザーがオブジェクトを複製したり、ウィンドウ再作成マクロを複数回実行したりすると、すべてのオブ

ジェクトがウェーブによってリンクされます。 

その後、そのうちの１つのオブジェクトを編集すると、その編集は関連するすべてのオブジェクトに影響し

ます。 

これは、この方式の利点とも考えられます。 

リテラル方式の欠点は、Igor がリテラルメソッドオブジェクトを含むウィンドウの再作成マクロを作成す

ると、すべての頂点またはコントロールポイントをテキストで指定しなければならないことです。 

これにより、作成と実行に多大な時間を要する巨大なマクロが作成される可能性があります。 

Igor は Experimentの保存と復元時に再作成マクロ技術を使うため、大きなリテラル方式オブジェクトを

使うと、Experimentの保存と復元にかかる時間が大幅に長くなる可能性があります。 

ウェーブ方式のオブジェクトには、この欠点はありません。 

ウェーブ方式の優れた機能の１つは、ユーザーがオブジェクトを編集した後でも、頂点やコントロールポイ

ントを読み戻すことができることです。 

ウェーブ方式のもう１つの利点は、新しい頂点やコントロールポイントをいつでも計算でき、依存するオブ

ジェクトが自動的に更新されることです。 

 画面の表示 

最初のポリゴンの頂点の値は、画面上の最初の頂点の位置を決定するものではないことに注意してくださ

い。 

位置は xOrg、yOrg パラメーターで指定されます。 

実際には、最初の頂点の値がすべての頂点から差し引かれ、その後、原点の値がすべての頂点に加算されま

す。 

従って、以下のコマンドは画面上に同じ表示を生成します： 

DrawPoly 120,50,1,1,{0,0,20,40,60,15} 

DrawPoly 120,50,1,1,{200,300,220,340,260,315} 

プログラミングでは、最初の頂点は通常 0,0 です。 

hScaling、vScaling パラメーターは、プログラマーにとってはあまり意味がないでしょう。 

両方の値には 1 を使ってください。 

GraphWaveDraw、GraphWaveEdit、GraphNormal 

これらのコマンドは、描画とは直接関係のないグラフモードに関連するものです。 

⚫ GraphWaveDraw：ポリゴン描画と同じユーザーインターフェイスを使って、ウェーブを描画できるモードに

グラフを切り替えます。 

⚫ GraphWaveEdit：ポリゴン編集と同じユーザーインターフェイスを使って、ウェーブを編集することができま

す。 

⚫ GraphNormal：ユーザーを通常操作モードに切り替えます。これはツールパレットの上部アイコンをクリック

する操作と同一です。 

これらのコマンドは、プログラムがユーザーにグラフ内に領域をスケッチさせることを可能にするために提供されて

います。 

プログラムは、ユーザーが行った操作を読み戻すことができます。 

他の描画モードとは異なり、これらのウェーブ描画および編集モードでは、ユーザー定義のボタンをアクティブにす

ることができます。 

これは、ユーザーに「完了」ボタンを提供するためです。 

ボタンプロシージャは、描画モードまたは編集モードを終了するために GraphNormal を呼び出す必要がありま

す。 

GraphWaveEdit コマンドは、コマンドでウェーブを指定するかどうかによって動作が少し異なります。 

ウェーブを指定した場合、そのウェーブのみユーザーが編集可能になります。 

ユーザーにウェーブを選択させる場合、ユーザーはクリックして新しいトレースに切り替えることができます。 

 

描画プログラミングの戦略 

図面作成プログラムを構成する方法は、主に 3つの異なるアプローチがあります： 

⚫ 追加：1つまたは複数のレイヤーの内容を追加できます。 

⚫ レイヤーの置き換え：レイヤーの内容を置き換えることができます。 

⚫ グループの置き換え：名前付きグループの内容を置き換えることができます。 

 

レイヤーを置き換える方法 

この方法を使うと、比較的複雑な図面を完全にプログラムのコントロール下で維持することができます。 

例えば、新しい軸タイプや新しい表示方法を追加して Igor を拡張したり、まったく新しい種類のグラフを作成した

りする場合があります。 

例えば、Polar Graphs パッケージは、replace メソッドを利用しています。 

置き換え方法の鍵は、SetDrawLayer コマンドに /K フラグを指定して使うことです。 

このフラグは、指定されたレイヤーのすべての内容を削除します。 

レイヤーをクリアした後、全体の内容を再描画する必要があります。 

これを行うには、通常、描画を維持するために必要なすべての情報を保持するための何らかのデータ構造を維持する

必要があります。 

例えば、人工的に軸パッケージを作成する場合、Igor の軸変更ダイアログに表示されるようなユーザー設定を維持

する必要があります。 

多くの場合、データフォルダーにいくつかのグローバル変数またはウェーブを設定するだけで十分です。 

例として、WaveMetrics Procedures フォルダーにある Drawing Axes プロシージャファイルを参照してくださ

い。 

 

グループを置き換える方法 

SetDrawEnv gname キーワードで作成された名前付きグループでは、DrawAction を使って、そのグループを削除

したり、新しい描画コマンドの挿入位置を設定したりすることができます。 

例については、DrawAction コマンドを参照してください。 

 

追加する方法 

この方法を使って、簡単な図面の作成を自動化する便利な機能を提供します。 

ユーザーがプロシージャを実行したり、ボタンをクリックしたりすると、指定したレイヤーにある既存の描画オブジ

ェクトに、ドロップライン、キャリブレーションバー、シェーディング矩形などの小さな描画が追加されます。 

一般的に、図面はユーザーが手動で作成できるもので、変更したいと思う可能性があります。 

レイヤーを指定する必要がある場合は、必ずユーザーレイヤーを選択してください。 

多くの場合、描画レイヤーを設定する必要はありません。現在のレイヤーを使ってください。 

ただし、特定の状況ではレイヤーを設定する必要がある場合があります。 

シェーディングの矩形は、グラフのトレースの下に配置する必要があるため、特定のレイヤーに配置すべきオブジェ

クトの一例です。 

この場合、SetDrawLayer コマンドを使う場合は、現在のレイヤーを「SetDrawLayer UserTop」で元の位置に戻

す必要があります。 

追加する方法を使っている場合は、Prog レイヤーの使用を避けてください。 

これは、Prog レイヤーがプログラム制御下でレイヤー全体を置き換えるために設計されているためです。 

もう 1つの考慮点は、デフォルトの描画環境を設定するかどうかです。 

一般的には設定しない方がよく、代わりにユーザーに制御を委ねるべきです。これは具体的な目的によって異なりま

す。 

 

グループ化 

図を作成する時は、グループ化を使うことをおすすめします。 

これにより、ユーザーは図全体を選択してドラッグするだけで、図全体を移動させることができます。 

 

例：ドロップライン 

この例では、カーソル A で識別されるウェーブの特定のポイントから、下軸に線を追加するプロシージャを示しま

す。 

このプロシージャは、グラフに左軸と下軸があり、カーソル A がトレース上にあることを前提としています。 

わかりやすくするため、エラーチェックは省略しています。 

Menu "Graph" 

 "Add Drop Line", AddDropLine() 

End 

 Function AddDropLine() 

 Variable includeArrow = 2   // 1=No, 2=Yes 

 Prompt includeArrow,"Include arrow head?", popup "No;Yes" 

 DoPrompt "Add Drop Line", includeArrow 

 includeArrow -= 1    // 0=No, 1=Yes 

 Variable xCursorPositon = hcsr(A) 

 Variable yCursorPosition = vcsr(A) 

 GetAxis/Q left 

 Variable axisYPosition = V_min 

 SetDrawEnv xcoord=bottom, ycoord=left 

 if (includeArrow) 

  SetDrawEnv arrow=1, arrowlen=8, arrowfat=0.5 

 else 

  SetDrawEnv arrow=0 

 endif 

 DrawLine xCursorPositon, yCursorPosition, xCursorPositon, axisYPosition 

End 

